

G E O M E T R I A C O N I L L O G O

INTRODUZIONE AL LOGO


Nelle schede LOGO che ti verranno presentate apprenderai via via le tecniche che ti permetteranno di progettare attraverso il linguaggio LOGO disegni che poi verranno eseguiti dal computer.

La conoscenza del linguaggio LOGO ti permetterà inoltre di consolidare e ritrovare le caratteristiche delle figure geometriche che stai studiando.

Alla fine di questo lavoro sarai in grado di programmare ed istruire il computer affinché realizzi disegni geometrici anche piuttosto complessi, come il «maniero» che ti viene presentato in figura 22.

Il progetto di MANIERO verrà sviluppato nelle sue parti che ti verranno presentate man mano che si incrementano le tue conoscenze.

Fig. 22



Il gioco che ti viene presentato in questa scheda può essere considerato un avvio all'uso corretto delle istruzioni del linguaggio LOGO.

Vedrai come sia possibile, con tale linguaggio, «progettare» i disegni che verranno eseguiti sul computer da una «tartaruga» che nel suo percorso simula il tratto della penna.

Situazione

Un gioco a squadre è stato così concepito: partecipano due squadre ognuna costituita da due ragazzi, un **comandante** e un **esecutore**.

L'esecutore deve percorrere un tracciato disegnato sul pavimento del cortile composto da una spezzata (fig. 23); le spezzate partono da *A* e da *B* e arrivano entrambe in *P*, dove c'è il premio in palio.

L'esecutore è bendato e riceve gli ordini dal suo comandante che lo deve guidare in modo corretto al premio.

Il numero complessivo dei passi da compiere e di svolte da effettuare nei due percorsi è identico, in modo da dare alle due squadre uguali opportunità di vittoria.


Le istruzioni a disposizione del comandante per guidare l'esecutore al premio, sono di due tipi:

di **avanzamento**, quantificato in passi; ad esempio: avanti 2 passi;

di **orientamento**, quantificato in svolta, a destra o a sinistra, di 90° ; ad esempio: gira a destra di 90° .

Vediamo quali sono le istruzioni che devono impartire entrambi i comandanti, tenendo presente che tutti e due gli esecutori partono orientati nella direzione del primo tratto di percorso e che nel disegno ad ogni lato di quadrato corrisponde un passo.

Fig. 23

*percorso A*

AVANTI 4 PASSI
 GIRA DESTRA 90°
 AVANTI 2 PASSI
 GIRA SINISTRA 90°
 AVANTI 3 PASSI
 GIRA DESTRA 90°
 AVANTI 3 PASSI
 GIRA SINISTRA 90°
 AVANTI 1 PASSO
 GIRA DESTRA 90°
 AVANTI 5 PASSI
 PRENDI IL PREMIO

percorso B

AVANTI 3 PASSI
 GIRA SINISTRA 90°
 AVANTI 2 PASSI
 GIRA DESTRA 90°
 AVANTI 2 PASSI
 GIRA DESTRA 90°
 AVANTI 2 PASSI
 GIRA SINISTRA 90°
 AVANTI 3 PASSI
 GIRA SINISTRA 90°
 AVANTI 6 PASSI
 PRENDI IL PREMIO

La vittoria è affidata alle abilità sia di comando, sia di corretta esecuzione dell'istruzione ricevuta.

In questa situazione, come in altre, è di fondamentale importanza l'uso appropriato di istruzioni che siano perfettamente comprensibili all'esecutore e che non diano origine ad interpretazioni differenti. L'esecutore sarebbe stato in difficoltà se il comandante gli avesse detto: «avanti un po'» «girati,...» o altre istruzioni similmente ambigue. Tali espressioni risulterebbero di tipo soggettivo e quindi non avrebbero reso unico, cioè **univocamente interpretabile**, il significato dell'istruzione.

Ti sarai reso conto di quanto sia meticoloso il lavoro di smontare in parti elementari una comunicazione

«complessa» come quella relativa al percorso che l'esecutore deve compiere. Risulta necessario smontare in parti elementari una comunicazione ogni qualvolta si voglia comunicare un'azione «complessa» ad un esecutore con capacità limitate ma ben precise. Il bambino bendato diventa un esecutore limitato nella sua capacità visiva, ma non motoria; in quella precisa situazione il bambino diventa un esecutore in grado di eseguire gli ordini che gli vengono impartiti (infatti sa contare i passi, conosce il significato di «svolta a destra di 90 gradi» ...) ma non è in grado però di decidere da solo il percorso perché, essendo bendato, non è in grado di vederlo.

A seconda della situazione la comunicazione avviene secondo codici diversi, ogni volta riferiti al tipo particolare di esecutore. In ogni caso si stabilisce un preciso legame ovvero un preciso tipo di comunicazione per ogni coppia **emittente - ricevente**.


Al variare delle caratteristiche del ricevente deve necessariamente cambiare il tipo di comunicazione. Questo tipo di rigore nella comunicazione fra emittente e ricevente è tipico dell'informatica. Infatti un elaboratore elettronico deve essere «istruito» in modo dettagliato e con un linguaggio ben preciso affinché sia in grado di eseguire un determinato compito. La comunicazione con un computer avviene, cioè, secondo codici ben precisi, codici che devono essere rigorosamente rispettati.

ESEMPIO

Dato il seguente percorso sul foglio quadrettato (fig. 24), stabiliamo la sequenza di comandi da impartire ad un ipotetico esecutore per andare dal punto *A* al punto *B*, dove l'avanzamento è relativo al lato di quadretto. L'orientamento della partenza e dell'arrivo sono indicati dal senso delle frecce.

GIRA DESTRA 90°
 AVANTI 2
 GIRA SINISTRA 90°
 AVANTI 2
 GIRA DESTRA 90°
 AVANTI 3
 GIRA DESTRA 90°
 AVANTI 1
 GIRA SINISTRA 90°
 AVANTI 2
 GIRA SINISTRA 90°
 AVANTI 3
 GIRA SINISTRA 90°
 AVANTI 1

Fig. 24


**ESERCIZI**

1. Dati i seguenti percorsi sul foglio quadrettato (fig. 25), stabilisci la sequenza di comandi da impartire ad un esecutore per andare dal punto *A* al punto *B*.

2. Traccia il percorso a cui fa riferimento la seguente successione di comandi.

AVANTI 2
 GIRA SINISTRA 90°
 AVANTI 1
 GIRA DESTRA 90°
 AVANTI 3
 GIRA SINISTRA 90°
 AVANTI 4
 GIRA SINISTRA 90°
 AVANTI 2
 GIRA DESTRA 90°
 AVANTI 2
 GIRA SINISTRA 90°
 AVANTI 1


Fig. 25



TI PRESENTIAMO LA TARTARUGA DEL LOGO

Il «mondo» del LOGO ti permette di essere, d'ora in poi, il comandante che impartisce istruzioni ad un esecutore che è la tartaruga del LOGO. Non aspettarti di incontrare una tartaruga vera e propria! Quando entrerai nel mondo del LOGO vedrai comparire un triangolino posto al centro dello schermo che, in seguito alle tue istruzioni, lascerà una traccia del suo passaggio. Tale triangolino simboleggia la tartaruga del LOGO.

Il LOGO ti fornisce uno strumento per disegnare figure non più con matita e righello, ma avvalendoti di un esecutore (la tartaruga) che disegna per te, lasciando una traccia, eseguendo quanto tu le dici di fare, attraverso una sequenza di ordini ben precisi. Quando tu decidi di far eseguire un certo compito alla tartaruga ti metti in una particolare situazione di comunicazione:

emittente (tu stesso)  ricevente (tartaruga)

La tartaruga è in grado di riconoscere, e quindi di eseguire, istruzioni ben precise, dalle quali non si discosta; per ottenere i risultati che desideri devi quindi metterti nella situazione di conoscere quello che lei è in grado di interpretare ed eseguire. Visto che il «linguaggio» della tartaruga è «povero» rispetto al tuo, devi compiere tu lo sforzo di adeguare i tuoi messaggi al suo vocabolario ristretto.

Se questo non avviene la tartaruga non sarà in grado di riconoscere il messaggio quindi non lo eseguirà. Se invece tu le darai un ordine corretto perché rientra nel vocabolario della tartaruga, ma sbagliato perché non corrisponde al percorso desiderato, la tartaruga eseguirà comunque il comando, senza però soddisfare le tue aspettative.

Chissà perché una tartaruga? forse perché per dover dar retta a tutti deve avere una pazienza da ... tartaruga! All'origine, quando fu creato il LOGO, i comandi venivano eseguiti da un piccolo robot a forma di tartaruga che si muoveva sul pavimento disegnando i suoi percorsi mediante una «penna» che poteva essere alzata o abbassata a seconda che la tartaruga dovesse lasciare la traccia o muoversi senza lasciare un segno.

Come far muovere la tartaruga

Quando entri in ambiente LOGO, sullo schermo compare un triangolino che simboleggia la tartaruga. La posizione «naturale» della tartaruga è al centro dello schermo, orientata verso l'alto (fig. 28).


Fig. 28



La prima istruzione di avanzamento

Un passo della tartaruga corrisponde ad un puntino luminoso lasciato sullo schermo. L'avanzamento di un passo è quindi uno spostamento molto piccolo. Quanti passi può fare la tartaruga a partire dalla posizione centrale per arrivare ai limiti dello schermo?

Fig. 29




I passi della tartaruga sono 320 in orizzontale e 250 in verticale (fig. 29). Attenzione quindi al numero di passi che le chiedi di compiere: se la tartaruga nel suo movimento supera questi limiti, solitamente scompare da un lato dello schermo per ricomparire dalla parte opposta.

Vediamo con alcuni esempi i primi comandi che ci consentono di far muovere la tartaruga sullo schermo.

Osserva nelle figure a pagina seguente l'effetto del comando impartito alla tartaruga quando si trova nella posizione naturale.


Fig. 30



AVANTI 50 (oppure A 50) (fig. 30).

↑
è necessario
uno spazio vuoto

Fig. 31



INDIETRO 50 (oppure I 50) (fig. 31).

Le istruzioni di avanzamento sono pertanto:

AVANTI (abbreviato **A**) $n \rightarrow$ avanza di n passi

INDIETRO (abbreviato **I**) $n \rightarrow$ arretra di n passi

Le prime istruzioni di orientamento

Abbiamo detto che la tartaruga è nella posizione naturale quando è orientata verticalmente e rivolta verso l'alto. Per orientare in modo diverso la tartaruga, è necessario


impartirle comandi che le facciano compiere una rotazione di ampiezza desiderata.

I comandi di orientamento sono seguiti da un numero che esprime l'ampiezza, in gradi, dell'angolo di cui si vuole far ruotare la direzione della tartaruga.

Vediamo con alcuni esempi i primi comandi che ci consentono di far ruotare la tartaruga rispetto alla posizione in cui si trova; osserva nelle figure l'effetto del comando impartito alla tartaruga quando si trova nella posizione naturale.


DESTRA 45 (oppure D 45) (fig. 32).

Fig. 32



SINISTRA 45 (oppure S 45) (fig. 33).

Fig. 33



Pertanto le istruzioni di orientamento sono:


DESTRA (abbreviato **D**) $n \rightarrow$ fa ruotare la tartaruga di un angolo di n gradi, in senso orario.

SINISTRA (abbreviato **S**) $n \rightarrow$ fa ruotare la tartaruga di un angolo di n gradi, in senso antiorario.

ESEMPI

1. Scriviamo i comandi necessari per far percorrere alla tartaruga la spezzata in fig. 34 partendo da A.

Fig. 34




```
D 90
A 20
S 90
A 20
D 90
A 30
D 90
A 10
S 90
A 20
S 90
A 30
S 90
A 10
D 90
```

L'ultimo comando serve per riportare la tartaruga nella posizione verticale.

È buona norma infatti riportare sempre, alla fine di un percorso, la tartaruga nella posizione verticale, rivolta verso l'alto.

2. Scriviamo i comandi necessari per far percorrere alla tartaruga il tracciato di figura 35, partendo da A, tenendo presente che in questo caso le svolte non sono sempre ad angolo retto.


Fig. 35



```
A 10
D 90
A 20
S 45
A 20
D 45
A 30
S 60
A 40
D 60
A 50
D 70
A 30
S 120
A 20
```


In figura 36 viene sottolineata l'importanza della direzione che ha già la tartaruga in un tratto di percorso, in relazione a quella che deve assumere per seguire il tratto successivo. Ti rendi conto quindi dell'importanza che assume la conoscenza degli angoli per far procedere correttamente la tartaruga.

Fig. 36



ESERCIZI


1. Determina la sequenza di istruzioni da impartire alla tartaruga per ottenere i poligoni raffigurati in figura 37.



2. Scrivi la sequenza di istruzioni necessarie per scrivere le cifre «5» e «3», dopo averle disegnate su un foglio quadrettato.

Due nuove istruzioni

Vediamo ora alcune nuove istruzioni per la tartaruga; supponiamo che debba disegnare la cifra 0 (zero; fig. 38).



La tartaruga non incontra difficoltà a compiere il percorso esterno; per tracciare la linea interna, dovrà percorrere dei tratti senza lasciare il segno del suo passaggio.

Per far questo la tartaruga dovrà muoversi momentaneamente *senza disegnare*. Sarà quindi necessario ordinarle di non lasciare traccia. Ricordiamo che in origine la tartaruga mecca-

nica aveva a disposizione una penna che poteva essere all'occorrenza alzata o abbassata, come d'altra parte fai tu quando lasci gli spazi fra una parola e l'altra.

I comandi che permettono di ottenere quanto detto sono:

SU → alza la penna.

GIU → abbassa la penna.

Dopo il comando SU la tartaruga esegue tutte le istruzioni successive muovendosi senza lasciare la traccia; per far sì che la tartaruga ricominci a disegnare è necessario dare l'istruzione GIU.

Vediamo le istruzioni da impartire alla tartaruga per scrivere la cifra «0», di fig. 38, partendo da A.

A 50 D 90

A 30 D 90

A 50 D 90

A 30 D 90

SU A 10 D 90 A 10 S 90 GIU

questa riga di istruzioni rappresenta la parte tratteggiata in rosso in figura.

A 30 D 90

A 10 D 90

A 30 D 90

A 10 D 90

3. Scrivi le istruzioni necessarie per realizzare i disegni nelle figure 41, 42, 43, 44 e 45.

Fig. 41





Fig. 42




Fig. 43




Fig. 44





Fig. 45



4. Stabilisci quale figura si genera eseguendo le seguenti istruzioni:

```
A 80 D 45 A 14.1 S 45 A 30 D 90 A 16 D 90
A 30 S 45 A 14.1 D 45 A 80 D 90 A 36
D 90 SU A 30 D 90 A 30 GIU S 90 A 20
D 90 A 30 D 90 A 20 D 90 A 30 D 90
```

Esiste inoltre la possibilità di eliminare la presenza del triangolino dallo schermo ordinando alla tartaruga di *nascondersi*; quando è utile poi controllare la sua posizione è necessario ordinarle di *riapparire*.

È inoltre possibile «ordinare» alla tartaruga di tornare nella posizione naturale ogniqualvolta si ritenga necessario mandarla da una posizione qualsiasi al centro dello schermo.

Le istruzioni che permettono di ottenere quanto detto sono:

NASTARTA (abbreviato **NT**) → nascondi la tartaruga

MOSTARTA (abbreviato **MT**) → mostra la tartaruga

TANA → riporta la tartaruga da un punto qualsiasi dello schermo nella posizione naturale.

G E O M E T R I A C O N I L L O G O

AVVIO ALLE PROCEDURE

In questa scheda vedremo come sia possibile raggruppare alcune istruzioni che devono essere ripetute più volte per eseguire un certo tracciato.

Ci proponiamo di «istruire» la tartaruga affinché esegua il seguente percorso (fig. 19).

Fig. 19



Scriviamo le istruzioni necessarie per eseguirlo:

```
D 90
A 10 SU A 5 GIU A 1 SU A 5 GIU
A 10 SU A 5 GIU A 1 SU A 5 GIU
A 10 SU A 5 GIU A 1 SU A 5 GIU
S 90
```

Come puoi osservare le istruzioni racchiuse in parentesi si ripetono identicamente tre volte; cosa che peraltro ti potevi aspettare osservando il ripetersi del «modulo» in colore in fig. 19. La tartaruga è in grado di ripetere un blocco di istruzioni un numero di volte da te stabilito. L'istruzione che ci consente di fare questo è:

RIPETI N [ISTRUZIONI]

che ripete le istruzioni specificate in parentesi quadra un numero n di volte.

Rivediamo le istruzioni precedenti utilizzando l'istruzione «RIPETI»:

```
D 90
RIPETI 3 [A 10 SU A 5 GIU A 1
SU A 5 GIU]
S 90
```

Come vedi, l'istruzione «RIPETI» ti consente di abbreviare notevolmente la sequenza di ordini da impartire alla tartaruga.


L'utilizzo di tale istruzione è comunque subordinato alla corretta individuazione del «modulo», che si ripete esattamente n volte nella figura. A volte, come in questo caso, restano all'esterno del blocco «RIPETI» istruzioni

che devono essere eseguite una sola volta, perché non fanno parte del modulo.

ESERCIZIO

Individua il modulo nei seguenti disegni (fig. 20) e scrivi le istruzioni necessarie per realizzarlo in LOGO.

Fig. 20



IL LOGO E IL TRIANGOLO EQUILATERO


In questo capitolo hai approfondito le tue conoscenze sui triangoli; hai appreso che un triangolo è equilatero quando ha angoli e lati congruenti. Vediamo le istruzioni da impartire alla tartaruga per realizzare sullo schermo un triangolo equilatero disposto come in fig. 21 (nella pagina seguente).

```
{ A 30 D 120
  A 30 D 120
  A 30 D 120
```


Le parentesi quadre raccolgono il modulo che a sua volta contiene il triangolo e lo spostamento.

Vediamo ora di utilizzare il triangolo in fig. 22 per una nuova «greca» (fig. 24).

Fig. 24



La sequenza di istruzioni necessarie per realizzare il triangolo, è la seguente:

```
D 30 RIPETI 3 [A 20 D 120] S 30
```

La tartaruga deve compiere lo spostamento di 10 passi fra un triangolo e il successivo; in questo caso non è necessario utilizzare le istruzioni «SU» e «GIU» in quanto può percorrere lo stesso tratto più di una volta senza modificarne la natura.

Le istruzioni quindi per il modulo completo sono le seguenti:

```
D 30 RIPETI 3 [A 20 D 120] S 30 D 90 A 10 S 90
┌──────────────────────────────────┐
└────────── TRIANGOLO ───────────┘ └── SPOSTAMENTO ───┘
└──────────────────────────────────┘
MODULO
```


Le istruzioni per realizzare l'intera «greca» sono pertanto:

```
RIPETI 6 [D 30 RIPETI 3 [A 20 D 120]
S 30 SUD 90 A 10 S 90 GIU]
```

ESERCIZIO

Scrivi la sequenza di istruzioni necessarie per disegnare la seguente greca (fig. 25).

Fig. 25



LE PROCEDURE IN LOGO


Il LOGO ti consente di denominare con un unico nome blocchi di istruzioni che vengono scritte per eseguire un certo lavoro. Tali «blocchi» prendono il nome di **procedure** e ad ognuna viene attribuito un nome che la identifichi.

Ogni procedura è composta da un inizio e una fine contraddistinti dalle parole chiave del linguaggio «PER» e «FINE».

Vediamo come trasformare la sequenza di istruzioni che hai già scritto per disegnare il numero «1», in una procedura a cui attribuiamo il nome «UNO» (fig. 26):

```
PER UNO
A 40 S 135
A 14 D 135
A 10 D 45
A 14 D 45
A 10 D 90
A 50 D 90
A 10 D 90
FINE
```

Fig. 26



Ogni procedura è composta da:

- un **titolo** che è costituito dalla parola chiave «PER» seguita da un nome che identifica la procedura;
- un **corpo** che è la sequenza di istruzioni della procedura;
- una **fine** costituita dalla parola chiave «FINE».

Per definire una procedura quindi:

PER nome una procedura deve iniziare sempre con la parola chiave «PER» seguita dal nome con cui vogliamo denominarla

istruzione 1
istruzione 2
...
istruzione n

FINE una procedura deve sempre terminare con la parola chiave «FINE».

Osservazione: non è necessario terminare ogni istruzione con il tasto di immissione (ENTER). È invece indispensabile l'utilizzo di tale tasto dopo la riga del «titolo» (la prima) e dopo quella che precede immediatamente l'istruzione «FINE»

ESEMPIO

Scriviamo la procedura per realizzare la «GIRANDOLA» di fig. 27.

```
PER GIRANDOLA
RIPETI 4 [A 20 D 60 A 25 D 120 A 20 S 90]
FINE
```

Fig. 27



Vediamo ora come definire, sotto forma di procedura, le istruzioni elencate per le figure 28, 29, 30 e 31 denominandole TRI-EQ1, TRI-EQ2, GRE-TRI1, GRE-TRI2.

La scrittura orizzontale o verticale delle istruzioni non ha alcuna conseguenza sull'esecuzione; la scelta di una

forma piuttosto dell'altra è quindi semplicemente dettata da considerazioni di praticità e leggibilità.

```
PER TRI-EQ1
RIPETI 3 [A 30 D 120]
FINE
```

Fig. 28



```
PER TRI-EQ2
D 30
RIPETI 3 [A 30 D 120]
S 30
FINE
```

Fig. 29




```
PER GRE-TRI1
RIPETI 5 [RIPETI 3 [A 20 D 120] SU
D 90 A 20 S 90 GIU]
FINE
```

Fig. 30



```
PER GRE-TRI2
RIPETI 6 [D 30 RIPETI 3 [A 20 D 120]
S 30 D 90 A 10 S 90]
FINE
```

Fig. 31



ESERCIZI

1. Individua il modulo del disegno rappresentato in fig. 32 qui a fianco e scrivi la procedura per realizzarlo utilizzando l'istruzione «RIPETI».

2. Individua il modulo in ciascuna delle «greche» nelle figure 33, 34 e 35; realizza la procedura utilizzando l'istruzione «RIPETI».

Fig. 32





Fig. 33



Fig. 34



Fig. 35



3. Scrivi la procedura per realizzare i disegni rappresentati nelle figure 36 e 37.

Fig. 36





Fig. 37




G E O M E T R I A C O N I L L O G O

PROCEDURE E SOTTOPROCEDURE

Il LOGO è in grado di arricchire il suo bagaglio di conoscenze di nuovi comandi; dopo aver definito una procedura questa può essere utilizzata all'interno di qualsiasi altra procedura.


Vediamo come, realizzando il programma che disegna la seguente farfalla (fig. 24).

Fig. 24



Puoi osservare che l'intero disegno è composto di tre parti distinte:

— ala sinistra; — corpo; — ala destra.



Scriviamo la procedura di ciascuna parte. Abbiamo già visto come si definisce la procedura triangolo equilatero che riutilizziamo per disegnare le ali. L'esecuzione prende avvio dal punto centrale.

```

PER ALA-D
D 60
RIPETI 3 [A 30 D 120]
S 60
FINE
  
```

```

PER ALA-S
S 60
RIPETI 3 [A 30 S 120]
D 60
FINE
  
```

Osserva che la procedura ALA-S differisce dalla procedura ALA-D esclusivamente per la sostituzione del comando «DESTRA» col comando «SINISTRA».

```

PER CORPO
A 15 D 45 A 5 I 5 S 90
A 5 I 5 D 45 I 45
FINE
  
```

La tartaruga ripercorre a volte tratti di strada già percorsi; questo tuttavia non modifica il risultato.

Invece di richiamare le tre procedure separatamente per far sì che la tartaruga esegua la farfalla, è possibile riunirle in un'unica procedura che chiameremo «FARFALLA» e che utilizza al suo interno le procedure ALA-D e ALA-S e CORPO. Tali procedure diventano **sotto-procedure** di «FARFALLA».

```

PER FARFALLA
ALA-D
ALA-S
CORPO
FINE
  
```

L'ordine con cui vengono richiamate le sottoprocedure «ALA-D», «ALA-S» e «CORPO» non è importante poiché, in questo caso, in ognuna di esse la tartaruga ha come punto di partenza il centro dello schermo.

Le procedure si possono utilizzare come fossero semplici istruzioni, all'interno dell'istruzione «RIPETI»; vediamo a questo proposito la realizzazione di «VENTOLA» (fig. 25 a pagina seguente). Per realizzarla utilizziamo la procedura del triangolo (TRI-EQU1), scritta nel seguente modo:

```

PER TRI-EQU1
RIPETI 3 [A 30 D 120]
FINE
  
```


```

PER VENTOLA
RIPETI 4 [TRI-EQU1 D 90]
FINE
  
```

La ventola che vogliamo realizzare è costituita da quattro pale distribuite a uguale distanza intorno al perno; poiché l'angolo giro misura 360° , ognuna dovrà essere distanziata dalla successiva di un angolo di 90° perché:

$$360^\circ : 4 = 90^\circ.$$

Fig. 25



IMPARIAMO A PROGRAMMARE

Così come nella realtà progettare significa analizzare e studiare le parti che concorrono alla realizzazione di un dato progetto, così in LOGO «progettare» significa scomporre ogni disegno nelle sue parti per realizzarle singolarmente.

Nella scheda LOGO a pag. 365 ci siamo proposti di «progettare» il disegno di «MANIERO» (fig. 26).

Alla luce delle conoscenze di LOGO che hai acquisito, ti puoi rendere conto che la realizzazione del progetto MANIERO può essere scomposta nella realizzazione distinta di alcune parti, secondo un modello gerarchico come in fig. 27 a pagina seguente.

Questa modalità di lavoro rientra in una logica tipicamente informatica che viene definita *top-down*, due parole inglesi che significano «dall'alto verso il basso». Il metodo *top-down* prevede la scomposizione in parti via via più semplici di un «progetto» complesso. Per schematizzare il progetto si utilizza una struttura ad albero come in fig. 27, in cui le parti più semplici si dispongono ai livelli inferiori, rispetto a quelle più complesse.

Fig. 26





Fig. 27



Abbiamo visto nel cap. 6 come il *top-down* costituisca un metodo di analisi anche per la risoluzione di problemi aritmetici.

Utilizzando tale procedimento, la realizzazione del progetto «MANIERO» può essere affidata a due o più gruppi di lavoro che ne realizzano ciascuno una parte. Tali parti verranno poi assemblate in un'unica procedura che realizzerà MANIERO.

Vediamo come ripartire il lavoro se si costituiscono due gruppi:

- *gruppo A*: si impegna a realizzare la procedura TORRE;
- *gruppo B*: si impegna a realizzare la procedura CORPO-MA (corpo del maniero).

In ogni gruppo poi vengono ulteriormente divisi i compiti:

- nel *gruppo A* alcuni progetteranno «BANDIERA», altri «TETTO», altri «FINESTRE» e altri ancora «CORPO-TO» (corpo della torre);
- nel *gruppo B* alcuni progetteranno «PORTONE», altri «MERLATURA» e altri «FINESTRE».

Componendo alla fine il lavoro di ogni gruppo si realizzerà l'intero progetto «MANIERO».

Con le conoscenze finora acquisite siamo in grado di seguire il lavoro dei sottogruppi TETTO (fig. 28), BANDIERA (fig. 29) e MERLATURA (fig. 30) che hanno realizzato le seguenti procedure:

PER TETTO

```

D 30
RIPETI 3 [A 35 D 120]
S 30
FINE
  
```

PER BANDIERA

```

A 8
RIPETI 3 [A 8 D 120]
FINE
  
```

PER MERLATURA

```

D 90 A 5
RIPETI 4 [A 5 S 90 A 10 D 90 A 10 D 90
          A 10 S 90 A 5]
A 5 S 90
FINE
  
```

Fig. 28



Fig. 29





Fig. 30



ESERCIZI

1. Osserva lo schema *top-down* in fig. 31; scrivi la procedura per ciascuna parte (fig. 32) e definisci la procedura principale «TELEVISORE».

Fig. 31






Fig. 32



2. Scomponi la fig. 33 in parti e disegna lo schema *top-down*; scrivi la procedura per ciascuna parte e definisci la procedura principale «CANDELA».

Fig. 33



3. Stabilisci quale figura si genera eseguendo le seguenti procedure; qual è la procedura principale?

PER MATTONE
RIPETI 2[A7 D90 A28 D90]
FINE

PER FILA1
RIPETI 5[MATTONE D90 A28 S90]
SU S90 A140 D90 GIU
FINE

PER FILA2
RIPETI 4[MATTONE D90 A28 S90]
SU S90 A112 D90 GIU
FINE


PER SPOSTA1
A7 D90 A14 S90
FINE

PER SPOSTA2
A7 S90 A14 D90
FINE

PER MURO
RIPETI 5[FILA1 SPOSTA1 FILA2 SU
SPOSTA2 GIU]
FINE

4. Scomponi in parti il disegno in fig. 34 e disegna lo schema *top-down*; scrivi la procedura per ciascuna parte e definisci la procedura principale.

Fig. 34




G E O M E T R I A C O N I L L O G O

PROCEDURE PER LA COSTRUZIONE DI QUADRILATERI


In questa scheda imparerai a definire alcune procedure relative ai quadrilateri studiati in questo capitolo. Definiamo la procedura «QUADRATO», che genera ad esempio un quadrato di lato 40 (fig. 29). Teniamo presente che il quadrato si ottiene ripetendo quattro volte un gruppo di due istruzioni: una di avanzamento e una di rotazione di 90° :

```
PER QUADRATO
RIPETI 4 [A 40 D 90]
FINE
```




Con la stessa logica procediamo alla definizione della procedura «RETTANGOLO» di dimensioni 30 e 50 (fig. 30):

```
PER RETTANGOLO
RIPETI 2 [A 30 D 90 A 50 D 90]
FINE
```




Definiamo ora la procedura per ottenere un «ROMBO» di lato 40 e con un angolo di 70° (fig. 31a):

```
PER ROMBO
RIPETI 2 [A 40 D 70 A 40 D 110]
FINE
```



Prima di iniziare il rombo in fig. 31b, la tartaruga dovrà compiere una rotazione di 35° .

```
PER ROMBO
S 35
RIPETI 2 [A 40 D 70 A 40 D 110]
D 35
FINE
```



Definiamo la procedura per ottenere un parallelogramma di lati 80 e 50, e con un angolo ampio 55° (fig. 32):

PER PARALLELOGRAMMA


D 35

RIPETI 2 [A 50 D 55 A 80 D 125]

S 35

FINE

Fig. 32



Infine vediamo la procedura per realizzare un trapezio isoscele (fig. 33):

PER TRAPEZIO

D 30 A 30

D 60 A 50


D 60 A 30

D 120 A 80

D 90

FINE

Fig. 33




PROCEDURE CON ARGOMENTO VARIABILE NEL LOGO


Vedrai ora come il LOGO faciliti ulteriormente il tuo lavoro.

Osserva il disegno in fig. 34; noterai che la forma predominante è il quadrato presente in diverse dimensioni.

Fig. 34



Lo schema ad albero delle procedure è il seguente:



Per realizzare la figura, con le attuali conoscenze, dovremmo definire almeno tre procedure di quadrati di dimensioni diverse, in cui varia ogni volta la misura del lato.

Questo comporterebbe un lungo e inutile lavoro; vediamo come il LOGO ci faciliti il compito consentendoci, con un'unica procedura, di tracciare quadrati diversi.

Per far questo è necessario non operare più direttamente con dei numeri, ma con «qualcosa» che li sostituisca.

Il LOGO permette di sostituire al valore del lato, che nella figura del «robot» varia di volta in volta, un nome, un'etichetta che rappresenti tutti i possibili valori numerici che ad essa andranno poi attribuiti.

Vediamo come: attribuiamo alla dimensione del lato del quadrato il «nome» LATO.


PER QUADRATO :LATO

RIPETI 4 [A :LATO D 90]

FINE

Puoi osservare che nella procedura quadrato compare il termine «LATO» sia nel titolo sia nella procedura; nel primo caso avverte che nella procedura verrà utilizzata l'etichetta «LATO» al posto dei dati numerici. Nel secondo caso l'etichetta «LATO» sostituisce il dato numerico che dovrebbe seguire il comando «AVANTI». Durante l'esecuzione della procedura verrà assegnato un preciso valore all'etichetta «LATO» e tale valore resterà associato ad ogni ricorrenza dell'etichetta stessa.

Fig. 35



In questo modo è stata scritta una procedura con **argomento variabile**; questo significa scrivere la procedura senza decidere immediatamente quale dimensione attribuire al lato. Quando si decide di utilizzare all'interno di una procedura un argomento variabile, è necessario specificarlo nel titolo accanto al nome della procedura. Come puoi osservare, infatti, l'etichetta «LATO» compare sia nella riga di intestazione sia nel corpo della procedura, preceduta da « : ». In generale quando si richiama un argomento variabile all'interno di una procedura, questo deve essere preceduto dal carattere « : ».

La procedura «QUADRATO» risulta così generalizzata e può essere utilizzata in circostanze diverse variando di volta in volta la scelta numerica dell'argomento «LATO». Ogni volta che vogliamo ottenere un quadrato particolare, dovremo richiamare la procedura «QUADRATO» facendola seguire dal valore numerico che è stato stabilito come misura di quel lato.

Ad esempio:

QUADRATO 10

disegnerà un quadrato di lato 10,

QUADRATO 50

disegnerà un quadrato di lato 50, ecc.

Vediamo quindi come realizzare le procedure necessarie per ottenere ROBOT, dopo aver definito la generica procedura «QUADRATO»:

```
PER OCCHIO
  QUADRATO 5
  QUADRATO 10
  FINE
```

Fig. 36



```
PER CONTORNO
  QUADRATO 40
  FINE
```

Fig. 37



Le procedure «BOCCA» e «ANTENNE» sono di facile scrittura con le conoscenze fin qui acquisite:

```
PER BOCCA
  A 3
  I 3
  D 90
  A 20
  S 90
  A 3
  I 3
  FINE
```

Fig. 38



PER ANTENNE

D 45
A 30
I 30
S 90
A 30
I 30
D 45
FINE

Fig. 39



Per comporre l'immagine completa di ROBOT, senza che le varie parti si sovrappongano, è necessario comporre le sotto-procedure già definite con gli spostamenti relativi alle varie parti che andremo a sviluppare successivamente.

Ciascuno spostamento consente di portare la tartaruga dal punto in cui termina una sottoprocedura al punto di inizio della successiva.

PER ROBOT

CONTORNO

SPOSTA1 ← SPOSTAMENTO VERDE

OCCHIO

SPOSTA2 ← SPOSTAMENTO ROSSO

OCCHIO

SPOSTA3 ← SPOSTAMENTO BLU

BOCCA


SPOSTA4 ← SPOSTAMENTO GRIGIO

ANTENNE

FINE

Nella procedura, come vedi, compaiono delle sottoprocedure relative agli spostamenti necessari per collocare ogni singola parte nella sua giusta posizione. Tali sottoprocedure potrebbero essere definite, come elenco di istruzioni, all'interno della procedura «ROBOT»; tuttavia risulta più coerente con una corretta modalità di programmazione, definirle come procedure autonome da collocarsi opportunamente all'interno della procedura «ROBOT».

Fig. 40



Vediamole:

PER SPOSTA1

SU A 20 D 90 A 25 S 90 GIU
FINE

PER SPOSTA2

SU S 90 A 20 D 90 GIU
FINE

PER SPOSTA3

SU I 15 D 90 A 5 S 90 GIU
FINE

PER SPOSTA4

A 35 S 90 A 10 D 90 GIU
FINE

Osserva che in tutte le procedure di spostamento sono stati utilizzati i comandi SU e GIU affinché la tartaruga si muova senza «sporcare» il disegno.

Riassumiamo le modalità per descrivere una procedura con un argomento variabile:

PER nome :VAR1

istruzione 1

istruzione 2

...

istruzione n

FINE

Ogni volta che, all'interno del corpo della procedura viene utilizzato un argomento variabile, deve essere sempre preceduto dal carattere « : ».

PROCEDURE CON PIÙ ARGOMENTI VARIABILI


Nel caso del quadrato entra in gioco una sola grandezza variabile: la dimensione del lato. Tale dimensione costituisce l'argomento variabile della procedura stessa. Non sempre però la grandezza variabile è unica. Per definire un generico rettangolo, ad esempio, è necessario definire due dimensioni. La procedura «RETTANGOLO» dovrà quindi prevedere l'utilizzo di due argomenti variabili che verranno entrambi esplicitati nel titolo della procedura; vediamo come, indicando le dimensioni con DIM1 e DIM2:

```
PER RETTANGOLO :DIM1 :DIM2
  RIPETI 2 [A :DIM1 D 90 A :DIM2 D 90]
FINE
```

Per ottenere il rettangolo di dimensioni ad esempio 20 e 50 (fig. 41) è necessario richiamare la procedura accompagnata dai due valori numerici richiesti:

```
RETTANGOLO 20 50
```


Fig. 41



Per ottenere più rettangoli aventi un vertice in comune (fig. 42) è necessario richiamare più volte la procedura «RETTANGOLO», assegnandole ogni volta coppie di argomenti diversi:

```
RETTANGOLO 10 40
RETTANGOLO 20 50
RETTANGOLO 30 60
```

Fig. 42



Riassumiamo le modalità per descrivere una procedura con più argomenti variabili che indichiamo con VAR1, VAR2,....:

```
PER nome :VAR1 :VAR2 ...
```

```
  istruzione 1
```

```
  istruzione 2
```

```
  ...
```

```
  istruzione n
```

```
FINE
```

ALCUNE PROPRIETÀ DEI QUADRILATERI ALLA LUCE DEL LOGO

Il rettangolo con LOGO

La procedura «RETTANGOLO», precedentemente definita, può consentirti anche di tracciare quadrati quando le due dimensioni assumono lo stesso valore.


RETTANGOLO 30 40 : è un rettangolo di dimensioni 30 e 40;

RETTANGOLO 40 50 : è un rettangolo di dimensioni 40 e 50;

RETTANGOLO 40 40 : è un quadrato di lato 40.

La procedura «RETTANGOLO» include in sé la possibilità di eseguire quadrati; questo conferma quanto già visto mediante la rappresentazione con il diagramma di Eulero-Venn, relativamente all'inclusione dell'insieme dei quadrati nell'insieme dei rettangoli.

Fig. 43



Il quadrato è, quindi, un particolare rettangolo.

Il rombo con LOGO

Vediamo ora come generalizzare la procedura «ROMBO», vista precedentemente; vediamo quali e quanti argomenti variabili è necessario individuare per la procedura «ROMBO».


Le grandezze variabili sono: la misura del lato e l'ampiezza degli angoli. Come puoi osservare in fig. 44, gli angoli in questione sono supplementari. Basterà quindi fissarne uno per conoscerli di conseguenza tutti, compresi quelli esterni che sono quelli che maggiormente interessano il percorso della tartaruga. I due argomenti variabili, che chiameremo LATO e ANG, devono comparire nell'intestazione della procedura «ROMBO», per essere utilizzati poi all'interno del programma.

```
PER ROMBO :LATO :ANG
```

```
  RIPETI 2 [A :LATO D 180- :ANG
            A :LATO D :ANG]
```

FINE

Fig. 44



La procedura «ROMBO» appena definita consente anche di tracciare quadrati ogniqualvolta la dimensione ANG assume il valore 90.


ROMBO 30 120 è un rombo di lato 30 e con un angolo di 120°;

ROMBO 30 100 è un rombo di lato 30 e con un angolo di 100°;

ROMBO 30 90 è un quadrato di lato 30.

La procedura «ROMBO» include in sé la possibilità di ottenere quadrati; questo conferma quanto già visto con la rappresentazione mediante diagramma di Eulero-Venn, relativamente all'inclusione dell'insieme dei quadrati nell'insieme dei rombi.


Fig. 45



Il quadrato è, quindi, un particolare rombo.

Riassumendo le figg. 43 e 44 si ottiene il seguente diagramma di Eulero-Venn (fig. 46):

Fig. 46



Il parallelogramma con il LOGO


Vediamo ora come generalizzare la procedura «PARALLELOGRAMMA» vista precedentemente; vediamo quali e quanti argomenti variabili è necessario individuare per la procedura.

Le grandezze variabili nella procedura «PARALLELOGRAMMA» sono: le dimensioni dei lati e, come accadeva per il rombo, l'ampiezza di un solo angolo. Chiameremo i tre argomenti variabili DIM1, DIM2 e ANG.

```
PER PARALLELOGRAMMA :DIM1 :DIM2
  :ANG
```

```
  RIPETI 2 [A :DIM1 D 180- :ANG
            A :DIM2 D :ANG]
```

FINE



PARALLELOGRAMMA 30 40 120 : è un parallelogramma generico;

PARALLELOGRAMMA 40 40 150 } rombi;
PARALLELOGRAMMA 50 50 130 }


PARALLELOGRAMMA 50 40 90 } rettangoli;
PARALLELOGRAMMA 40 50 90 }

PARALLELOGRAMMA 40 40 90 } quadrati.
PARALLELOGRAMMA 50 50 90 }

In generale, quindi, la procedura «PARALLELOGRAMMA» consente di ottenere:

- **rombi**, quando DIM1 è uguale a DIM2;
- **rettangoli**, quando ANG assume il valore 90;
- **quadrati**, quando DIM1 coincide con DIM2 e inoltre ANG assume il valore 90.

Quanto visto conferma la rappresentazione mediante diagramma di Eulero-Venn relativa ai parallelogrammi (fig. 48).



Riprendiamo il progetto di maniero.

Per la progettazione di «MANIERO» il lavoro è stato suddiviso in due gruppi distinti che devono realizzare TORRE e CORPO-MA (corpo maniero); è importante stabilire per convenzione che entrambi abbiano come riferimento la posizione naturale della tartaruga (TANA). Ciascuno dei due gruppi quindi deve concludere il proprio lavoro con le istruzioni:

SU TANA GIU

che permettono alla tartaruga di tornare nella posizione naturale.

Con le conoscenze finora acquisite, possiamo seguire il lavoro del gruppo A che realizza la procedura «TORRE», di cui viene data la seguente scomposizione gerarchica, nella prossima scheda seguiremo il lavoro del gruppo B.

Fra le procedure evidenziate nell'albero gerarchico, «TETTO» e «BANDIERA» sono già state definite nella scheda LOGO di pag. 124.

Le rimanenti possono essere progettate utilizzando la procedura «RETTANGOLO» definita in questa scheda.


```
PER RETTANGOLO : DIM1 : DIM2
RIPETI 2 [A : DIM1 D 90 A : DIM2 D 90]
FINE
```

```
PER FINESTR2 (in blu in figura)
RIPETI 3 [RETTANGOLO 15 5 SU A 20
GIU]
```

```
FINE
PER FINESTR3
RIPETI 3 [RETTANGOLO 5 5 SU A 10
GIU]
```

```
FINE
PER CORPO-TO
RETTANGOLO 120 35
FINE
```


```
PER TORRE
CORPO-TO
A 50 D 90 SU A 15
S 90 GIU
FINESTR2
S 90 SU I 15 GIU
FINESTR3
D 90 A 10
TETTO
D 30 A 35 S 30
BANDIERA
SU TANA GIU
FINE
```



ESERCIZI

1. Definisci la procedura, con argomento variabile, «TRI-EQU» e scrivi la procedura «TENDA» utilizzando la procedura «TRI-EQU» (fig. 50). È possibile definire un'unica procedura di spostamento?


Fig. 50



2. Definisci la procedura, con argomento variabile, «QUADRATO»; scrivi la procedura «CANNOCCHIALE» utilizzando la procedura «QUADRATO» come sottoprocedura (fig. 51).

È possibile definire un'unica procedura di spostamento? Pensi che sia possibile risolvere il problema utilizzando un argomento variabile nella procedura di spostamento?

Fig. 51




3. Osserva il disegno in fig. 52 e la relativa procedura.

PER ESTREMO1

D 90 A 10 S 80 A 30

FINE

Fig. 52



PER SPIRA

D 160 A 60 S 160 A 60

FINE

PER ESTREMO2

D 160 A 30 S 80 A 10 S 90

FINE

PER MOLLA : N

ESTREMO1

RIPETI : N [SPIRA]

ESTREMO2

FINE

Modifica le procedure date in modo da rendere variabile sia l'angolo fra una spirale e la successiva sia la lunghezza delle spire.

4. Osserva le procedure qui di seguito riportate e stabilisci quale figura si genera eseguendole.

PER RETTANGOLO : L1 : L2

RIPETI 2[A:L1 D 90 A:L2 D 90]

FINE

PER

RETTANGOLO 20 80

A 20 D 90 A 10 S 90

RETTANGOLO 15 60

A 15 D 90 A 15 S 90

RETTANGOLO 10 30

A 10 D 90 A 5 S 90

RETTANGOLO 5 20

A 5 D 90 A 8 S 90

RETTANGOLO 18 4

NT

FINE

5. Scrivi le procedure necessarie per realizzare il disegno in fig. 53.

Fig. 53





Fig. 41



Come puoi osservare, nelle procedure precedenti varia, di volta in volta, il numero dei lati e di conseguenza l'ampiezza dell'angolo esterno. La procedura relativa ad un generico poligono regolare di lato 30 passi può quindi essere scritta nel seguente modo:

```
PER POL-RE30 :N
RIPETI :N [A 30 D 360/:N]
FINE
```

Questa procedura permette di eseguire un poligono regolare di lato 30 passi e numero di lati N fissato di volta in volta.

Esempio

POL-RE30 3 _ realizza un triangolo equilatero di lato 30 passi.

POL-RE30 4 _ realizza un quadrato di lato 30 passi.

Ci proponiamo ora di generalizzare al massimo la procedura del poligono regolare, rendendo variabile anche la misura del lato che indichiamo con L ; avremo quindi:


```
PER POL-REG :N :L
RIPETI :N [A :L D 360/:N]
FINE
```

Eseguiamo la procedura «POL_REG» facendo variare prima il numero dei lati tenendo fissa la misura (L), poi facendo variare la misura del lato tenendo però fisso il numero di lati (N) (fig. 42):

Vediamo ora come utilizzare la procedura «POL_REG» per realizzare i rosone di stile gotico che seguono.

Il primo rosone (fig. 43a) ha come figura base un ottagono di lato 15 passi che viene ripetuto 8 volte, il secondo un

Fig. 42




decagono di lato 15 che viene ripetuto 10 volte e l'ultimo un dodecagono, anch'esso di lato 15, ripetuto 12 volte. In ogni rosone la somma delle rotazioni del poligono «base», evidenziato in colore, è calcolata in modo tale da realizzare un giro completo di 360° . In ogni rosone viene pertanto ripetuta n volte la rotazione di $360^\circ/n$, esaurendo così l'intero angolo giro.

Vediamo le tre procedure per realizzare i rosone della figura 43a, 43b, 43c:


```
PER ROSONE1
RIPETI 8 [POL-REG 8 15 D 360/8]
FINE
```

Fig. 43a



```
PER ROSONE2
RIPETI 10 [POL-REG 10 15 D 360/10]
FINE
```

Fig. 43b



PER ROSONE3
 RIPETI 12 [POL-REG 12 15 D 360/12]
 FINE

↓ ↓ ↓ ↓
 N N L N

Osserviamo che in ciascuna procedura viene utilizzata «POL-REG» come sottoprocedura.

La procedura «ROSONE» può essere generalizzata in modo da poter decidere di volta in volta quale poligono regolare utilizzare e quale misura assegnare al lato:

PER ROSONE :N :L
 RIPETI :N [POL-REG :N :L D 360/:N]
 FINE

Vediamo ora la realizzazione di un rosone mediante la precedente procedura nel caso di un poligono «base» di 14 lati di 10 passi (fig. 44); occorrerà eseguire la procedura:

ROSONE 14 10

Fig. 43c





Fig. 44




LA CIRCONFERENZA CON LOGO

Ti puoi rendere conto, utilizzando la precedente procedura «POL-REG» con differenti argomenti, che ti è semplice ottenere poligoni regolari con un numero molto alto di lati. In figura 45 sono mostrati, ad esempio, poligoni regolari di 90 e di 360 lati con misura del lato ovviamente contenuta.

POL-REG 90 2 POL-REG 360 1
 POL-REG 360 0.3

Fig. 45




Puoi osservare che i poligoni ottenuti, all'aumentare del numero di lati, perdono sempre più le caratteristiche dei poligoni per acquisire quelle di un'altra figura geometrica che da essi viene approssimata: la circonferenza.

Potremo quindi utilizzare la procedura «POL_REG» ogniqualvolta sia necessario disegnare una circonferenza.

ESEMPIO

Utilizziamo la procedura «POL_REG» per ottenere le seguenti circonferenze tangenti internamente (fig. 46):


```
PER CIRC-TAN
POL-REG 90 1
POL REG 90 2
POL REG 90 3
FINE
```




Le conoscenze acquisite ci permettono ora di completare la progettazione di MANIERO seguendo il lavoro del gruppo B che realizza la procedura «CORPO MA» e le relative procedure (fig. 47):

```
PER CORPO-MA
PORTONE
SU S 90 A 67.5 D 90 A 45 GIU
MERLATURA
SU I 10 S 90 A 2.5 GIU
FINESTR1
SU TANA GIU
FINE
```

```
PER PORTONE
PORTA
SU S 30 A 17.5 D 90
A 10 S 90 GIU
GRATA
FINE
```




```
PER PORTA
RIPETI 2 [RIPETI 2 [RETTANGOLO 30
12.5] D 90 A 12.5 S 90]
D 90 A 10 I 90 A 55 S 90 A 12.5 D 90
SU S 5 S 60 GIU
TRIANGOLO 5
D 60 A 5
SU A 5 S 60 GIU
TRIANGOLO 5
FINE
```




```
PER GRATA
RIPETI 18 [A 2.2 S 10]
A 2 S 90 A 20.5 S 90
RIPETI 18 [A 1.4 S 10]
S 90 A 12 S 90
RIPETI 18 [A 0.6 S 10]
S 90 A 3
RIPETI 5 [S 30 A 11.5
I 11.5]
```

```
D 60
FINE
```



```
PER FINESTR1
RIPETI 9 [RETTANGOLO 5 5 SU A 10 GIU]
FINE
```

Le procedure specifiche di MANIERO utilizzano le procedure TRIANGOLO e RETTANGOLO già viste nelle schede precedenti:



PER RETTANGOLO : DIM1 : DIM2
RIPETI 2 [A : DIM1 D 90 A : DIM2 D 90]
FINE

PER TRIANGOLO : LATO
RIPETI 3 [A : LATO D 120]
FINE

A questo punto siamo in grado di definire la procedura di «MANIERO» che comprende le procedure per essa definite e gli spostamenti necessari per ottenere la corretta composizione della figura. Vediamo la procedura «MANIERO»:

PER MANIERO
PS
SPOSTA1
CORPO-MA
SPOSTA2
TORRE
SPOSTA3
TORRE
SU TANA GIU NT
FINE

Il comando «PS» (Pulisci Schermo) va messo all'inizio della procedura principale per «pulire» lo schermo da altri eventuali disegni, tracciati precedentemente

PER SPOSTA1
SU I 90 D 90 A 5 S 90 GIU
FINE

questa procedura permette di posizionare la tartaruga in modo che inizi l'esecuzione di «CORPO-MA» nel punto desiderato.


PER SPOSTA2
SU S 90 A 50 S 90 A 90 D 90
A 35 D 90 GIU
FINE

questa procedura permette di spostare la tartaruga da «TANA» all'inizio della torre di sinistra.

PER SPOSTA3
SU D 90 A 40 D 90 A 90
S 180 GIU
FINE

questa procedura permette di spostare la tartaruga da «TANA» all'inizio della torre di destra.

Rivediamo globalmente la progettazione di MANIERO, realizzata secondo lo schema ad albero già presentato, nel quale compaiono ora le singole procedure sviluppate:




G E O M E T R I A C O N I L L O G O

LA SIMMETRIA NEL LOGO

In questa scheda analizzeremo le tecniche necessarie per realizzare figure che si corrispondono in una simmetria. Prendiamo in esame alcune procedure già viste in precedenza e vediamo come, con semplici accorgimenti, sia possibile ottenere procedure che realizzino figure simmetriche alle date.

Consideriamo il motivo in fig. 37: è composto da due parti perfettamente simmetriche rispetto all'asse indicato.




Andiamo ad analizzare le procedure necessarie per realizzare la figura nelle sue due parti; chiamiamo GRE-SIN la parte di sinistra della greca e GRE-DES la parte destra:

PER GRE-SIN	PER GRE-DES
S 90 A 10	D 90 A 10
D 90 A 35	S 90 A 35
S 90 A 25	D 90 A 25
S 90 A 25	D 90 A 25
S 90 A 10	D 90 A 10
S 90 A 10	D 90 A 10
S 90 A 20	D 90 A 20
SU TANAGIU	SU TANAGIU
FINE	FINE

Le due procedure presentano, come puoi osservare, un'unica differenza: le istruzioni relative alla rotazione verso destra

Fig. 38



o verso sinistra risultano scambiate. In questo modo tutti gli spostamenti che avvenivano verso destra avvengono poi identicamente verso sinistra e viceversa (fig. 38). Possiamo quindi pensare che per ottenere una figura simmetrica rispetto ad una data, bisogna semplicemente scambiare fra loro, nella procedura, i comandi «DESTRA» e «SINISTRA».

La procedura che realizza l'intera greca è pertanto la seguente:


```
PER RICCIOLO
GRE-SIN
GRE-DES
FINE
```

ESERCIZIO

Modifica le seguenti procedure in modo tale che realizzino figure simmetriche e verifica al computer i risultati ottenuti:

PER QUAD-DES	PER QUAD-SIN
RIPETI 4 [A 40 D	
90]	...
FINE	...
PER TRI-DES	PER TRI-SIN
RIPETI 3 [A 40 D	
120]	...
FINE	...
PER PARAL-DES	PER PARAL-SIN
...	S 30
...	RIPETI 2 [A 20 S 60 A
...	40 S 120]
	D 30
	FINE

Vogliamo ora realizzare l'immagine in fig. 39, a pagina seguente, che ricorda gli occhi della civetta.



L'immagine, come puoi osservare, è simmetrica rispetto all'asse indicato; vediamo le procedure necessarie per realizzarla. A questo scopo decidiamo di utilizzare la procedura «POL-REG» studiata precedentemente, che realizza il poligono procedendo verso destra.


```
PER POL-REGD :N :L
RIPETI :N [A :L D 360 / :N]
FINE
```

La procedura «simmetrica» diventa:

```
PER POL-REGS :N :L
RIPETI :N [A :L S 360 / :N]
FINE
```

Scriviamo ora la procedura per realizzare gli occhi della civetta e la chiamiamo «CIVETTA».

```
PER CIVETTA
POL-REGD 8 15
POL-REGD 8 20
POL-REGD 8 25
POL-REGD 8 30
POL-REGD 8 35
POL-REGS 8 15
POL-REGS 8 20
POL-REGS 8 25
POL-REGS 8 30
POL-REGS 8 35
FINE
```



In questo modo abbiamo realizzato una serie di poligoni simmetrici; vogliamo anche realizzare il «riempimento» con colore della parte più interna dell'occhio. Per far questo devi incrementare la tua conoscenza del LOGO di nuove istruzioni.

Per riempire una zona racchiusa in una poligonale chiusa dovremo portare la tartaruga all'interno di tale zona e usare un'istruzione che ne consenta il «riempimento» con un dato colore. La tartaruga con l'istruzione «RIEMPI» procede al riempimento della zona in cui si trova, con il colore di cui è provvista in quel momento la penna. Per cambiare il colore è necessario dare l'istruzione «ASCOL» seguita dal numero del colore desiderato.

Volendo riempire con il colore di cui è provvista la penna gli occhi della civetta, le istruzioni da inserire nella procedura sono:

SUD 90 A 10 GIU → queste istruzioni servono per mandare la tartaruga nella zona da colorare senza lasciare la traccia.

RIEMPI

Le nuove istruzioni apprese nella scheda sono:

ASCOL n-colore	→	assegna il colore «n-colore» alla penna.
RIEMPI	→	riempie del colore assegnato alla penna lo spazio chiuso entro cui si trova la tartaruga.